

Basic Express Application Note

Using a Sharp GP2Y0A02YK Infrared Ranger with BasicX

Introduction

The Sharp GP2Y0A02YK infrared ranger is able to continuously measure the distance to an object. The usable range is 20 cm to 150 cm. The device generates an analog voltage that is a function of range, and the output voltage can be measured by an analog-to-digital (ADC) input line on a BasicX system.

Hardware connections

Figure 1 (below) illustrates the hardware interface between the detector and BasicX system:




Figure 1

A voltage between 4.5 V and 5.5 V is needed to power the detector. The current drain is approximately 33 mA to 50 mA. The device starts operating as soon as power is applied.

The analog voltage output is connected to one of the ADC input pins on the BasicX system. The 470 Ω resistor (shown above) is optional.

How it works

The device emits a pulsed infrared beam at a wavelength of $850 \text{ nm} \pm 70 \text{ nm}$. If an object is within range and in line with the IR beam, reflected light forms an image on a linear CCD array in the receiver. Triangulation is then used to determine range. Readings are updated at a rate of approximately 24 Hz.

The detector is relatively insensitive to ambient lighting, as well as reflectivity of the object being detected. It is possible for the device to detect relatively dark objects in full sunlight.

Analog Output Voltage vs. Distance to Reflective Object




Figure 2

The output voltage is a nonlinear function of the distance between the object and the receiver. The curve in figure 2 was taken from Sharp documentation. The curve assumes a white object with a 90 % reflectivity.

Note that objects closer than about 15 cm can look like objects at longer distances. This ambiguity needs to be taken into account if objects are allowed to be closer than the 15 cm threshold.




Figure 3

An curve fit formula is used to approximate distance as a function of voltage:

$$D = \frac{A + B X}{1 + C X + D X^2}$$

where

- D = Distance (cm)
- V = Voltage (V)
- A = 0.008 271
- B = 939.6
- C = -3.398
- D = 17.339

Software interface

To read the sensor, procedure GetADC is called in order to read the analog voltage generated by the device. GetADC returns nondimensional voltage, which needs to be multiplied by 5 to convert to units of volts. Function VoltageToRange then converts voltage to distance, where distance is in units of cm.

Source code is provided in a separate file called SharpGP2Y0A02YK.bas. A listing of the program is shown below.

The program transmits range data in ASCII format through the BasicX Com1 serial port. The port is configured to 19 200 baud, 8 data bits, 1 stop bit, no parity.

```
'-----  
Option Explicit  
  
' This program reads a Sharp GP2Y0A02YK infrared object detector, and  
' continuously displays range at a rate of about 3 times per second.  
'-----  
Public Sub Main()  
  
    Dim Range As Single  
    Dim Success As Boolean  
  
    Debug.Print  
    Debug.Print "Sharp GP2Y0A02YK IR object detector"  
    Debug.Print  
  
    Do  
        Call GetRange(Range, Success)  
  
        Debug.Print "Range = ";  
  
        If (Success) Then  
            Debug.Print CStr( CInt(Range) ); " cm"  
        Else  
            ' Out of range.  
            Debug.Print " ***"  
        End If  
  
        Delay 0.3  
    Loop  
  
End Sub  
'-----
```

```

'-----
Public Sub GetRange( _
    ByRef Range As Single, _
    ByRef Success As Boolean)

    Dim Voltage As Single
    Const MinVolt As Single = 0.4
    Const MaxVolt As Single = 2.8

    ' This pin number is for a BX-24. The pin number may need to be changed
    ' depending on the BasicX system being used.
    Const InputPin As Byte = 16

    Call GetADC(InputPin, Voltage)

    ' Convert to voltage.
    Voltage = Voltage * 5.0

    Range = VoltageToRange(Voltage)

    ' Check for legal voltage.
    If (Voltage >= MinVolt) And (Voltage <= MaxVolt) Then
        Success = True
    Else
        Success = False
    End If

End Sub

'-----
Private Function VoltageToRange( _
    ByVal V As Single) As Single

    ' Returns distance in units of cm.

    Const A As Single = 0.0082712905
    Const B As Single = 939.57652
    Const C As Single = -3.3978697
    Const D As Single = 17.339222

    ' Curve fit.
    VoltageToRange = (A + B * V) / (1! + C * V + D * V * V)

End Function

'-----

```

© 2002 by NetMedia, Inc. All rights reserved.

Basic Express, BasicX, BX-01, BX-24 and BX-35 are trademarks of NetMedia, Inc.

All other trademarks are the property of their respective owners.

2.01B